
argparse: command line parser for Erlang
    

[image: Build Status] [image: Hex.pm] [image: Hex Docs]
A mini-framework to create complex cli. Inspired by Python argparse.
Warning
This project is no longer maintained, because argparse is now a part of
Erlang/OTP: https://www.erlang.org/doc/man/argparse.html (starting with OTP 26).
To accommodate with this change, argparse module has been renamed to args
in version 2.0.0. Applications that need argparse for earlier OTP versions
can use this library until the lowest supported version is OTP 26, and then
just move forward to argparse provided by OTP.
There are minor differences in the command line options description between
args and OTP argparse:
	int type is renamed integer
	exceptions are accompanied by extended error information


Follows conventions of  Unix Utility Argument Syntax.
    argparse [-abcDxyz][-p arg][operand]

Argument parser
Converts list of strings (command line) into an argument map,and a
command path; see argparse reference for detailed description.
cli mini-framework
Make a step beyond parser, and export existing Erlang functions: cli reference
Basic example
cli is naturally suitable for building small escript-based apps:
    #!/usr/bin/env escript

    -export([main/1, cli/0, rm/1]).
    -behaviour(cli).
    -mode(compile). %% evaluated module cannot contain callbacks

    main(Args) ->
        cli:run(Args, #{progname => "simple"}).

    cli() ->
      #{
        handler => {?MODULE, rm},
        arguments => [
          #{name => force, short => $f, type => boolean, default => false},
          #{name => recursive, short => $r, type => boolean, default => false},
          #{name => dir}
      ]}.

    rm(#{force := Force, recursive := Recursive, dir := Dir}) ->
        io:format("Removing ~s (force ~s, recursive: ~s)~n",
            [Dir, Force, Recursive]).
The example above does not have sub-commands, and implements rm/1
handler, that serves as an entry point with parsed arguments. Help options are
added automatically:
    $ ./erm --help
    usage: erm [-fr] <dir>

    Optional arguments:
      dir
      -r  recursive, [false]
      -f  force, [false]

Calc: cli with multiple commands
Calculator implements several commands, with sub-commands available. Full
source code here: doc/examples/escript/calc
Command definitions:
    cli() ->
        #{
            commands => #{
                "sum" => #{
                    arguments => [
                        #{name => num, nargs => nonempty_list, type => int, help => "Numbers to sum"}
                    ]
                },
                "math" => #{
                    commands => #{
                        "sin" => #{handler => {math, sin, undefined}},
                        "cos" => #{},
                        "tan" => #{handler => {math, tan, undefined}}
                    },
                    arguments => [
                        #{name => in, type => float, help => "Input value"}
                    ]
                },
                "mul" => #{
                    arguments => [
                        #{name => left, type => int},
                        #{name => right, type => int}
                    ]
                }
            }
        }.
The calculator provides "sum" command that prints a sum of integer numbers:
    $ ./calc sum 1 2 3
    6

Math sub-commands provide trigonometric functions:
    $ ./calc math cos 1.4
    0.16996714290024104
    $ ./calc math sin 1.4
    0.9854497299884601

Complex applications
cli is capable of handling releases containing hundreds of modules
implementing cli behaviour. Commands may be exported from multiple modules and
applications. cli makes best efforts to merge commands exported,
format usage output and error messages.
See example: doc/examples/multi
This example contains two modules, multi_math.erl and multi_string.erl.
Use rebar3 escriptize to build the application. Try various commands,
e.g. ./multi math cos 1.0, or ./multi string lexemes 1+2+3+4 -s +
to get a feeling!
Argument parser alone
It is possible to use argument parser alone, without the cli mini-framework:
    #!/usr/bin/env escript

    main(Args) ->
        #{force := Force, recursive := Recursive, dir := Dir} =
            args:parse(Args, cli()),
        io:format("Removing ~s (force: ~s, recursive: ~s)~n",
            [Dir, Force, Recursive]).

    cli() ->
        #{arguments => [
            #{name => force, short => $f, type => boolean, default => false},
            #{name => recursive, short => $r, type => boolean, default => false},
            #{name => dir}
        ]}.
Help and usage information
cli automatically prints usage, if command line parser reports an
error. An attempt is made to guess the most relevant command.
Argument help can be customised.
Build
This project requires OTP-22 or above. Simple integration is available via Hex and
rebar3.
    {deps, [argparse]}.
Known incompatibilities
In comparison with Python implementation, argparse for Erlang:
	boolean flag (option) automatically uses {store, true}
	all positional arguments are required by default (even when nargs is 'maybe')
	first-class (sub) commands, slightly differently from argparse
	implicit --help/-h is not a part of argparse (but implemented in cli)

Commands vs. positional arguments: command always takes precedence
over a positional argument.
Commands form exclusive groups, e.g. only one command can
be followed at a time.
Supported command line syntax
	command (priority positional argument) : ectl {crawler|reader|writer}
	command, and sub-command:                ectl crawler {start|stop|check}
	positional argument (required):          ectl <arg1> <arg2>
	positional argument (with default):      ectl [<arg1>]
	boolean flag:              ectl [-rf]
	required flag:             ectl -r
	short optional argument:   ectl [-i <int>]
	short optional:            ectl [-i [<int>]]
	required short option:     ectl -i <int>
	long option flag:          ectl [--foo]
	long optional argument:    ectl [--foo <arg>]
	required long:             ectl --foo <arg>
	long, arg=value form:      ectl --foo=arg
	list of arguments:         ectl <arg>, ...

Expected features
To be considered after 1.2.0:
	search for commands and arguments (mini-man)
	abbreviated long forms
	mutual exclusion groups
	shell auto-complete
	automatically generated negative boolean long forms "--no-XXXX"




  

    
cli mini-framework
    

cli is designed to simplify command-line interface integration. From a tiny escript,
to a system exporting several hundred commands.
Basic example
Ensure argparse is in your code path when running examples.
    cd doc/examples/escript
    ERL_FLAGS="-pa ../../../_build/default/lib/argparse/ebin" ./calc mul 2 2

Implementing a utility with a single command to run requires:
	Compiled code (as evaluated code does not work with callbacks)

	'cli' behaviour declared

	cli/0 callback returning arguments map and a handler
 #!/usr/bin/env escript
 -export([main/1, cli/0, cli/1]).
 -behaviour(cli).
 -mode(compile). %% evaluated module cannot contain callbacks
 main(Args) ->
 cli:run(Args, #{progname => "simple"}).
 cli() ->
 #{arguments => [
     #{name => force, short => $f, type => boolean, default => false},
     #{name => recursive, short => $r, type => boolean, default => false},
     #{name => dir}
 ]}.
 cli(#{force := Force, recursive := Recursive, dir := Dir}) ->
 io:format("Removing ~s (force ~s, recursive: ~s)~n",
     [Dir, Force, Recursive]).


rebar3 escript example
Creating a new application exposing CLI is as simple as:
	Running rebar3 new escript conf_reader
	Adding argparse to deps and escript_incl_apps in rebar.config
	Add a function (cli/0) declaring CLI arguments
	Use the specification: args:parse(Args, cli())
	Run rebar3 escriptize to build the application.

Command-line interface discovery
By default, cli:run/1 scans all loaded modules to find those implementing
cli behaviour.
Use run/2 with modules option to specify a single module, or a
list of modules to search (default is all_loaded):
cli:run(["arg"], #{modules => ?MODULE}).
When there are multiple modules in the list, cli merges all arguments and
commands exported by these modules. In case of a conflict (e.g. clashing
short or long option name, or top-level command name), first match is
accepted, and all others are discarded (with warning emitted to OTP logger,
can be switch off with warn flag set to false).
Be careful with top-level arguments exported: they are added to
all commands produced by all modules - this side effect is usually
undesired. There is a warning emitted if more than one module implements
cli behaviour, and global arguments are exported.
Handler specification
Handler is a callback invoked by cli:run when command line parser completes successfully.
Handler may accept a single argument (map of argument names to their values, as returned
by argparse):
sum(#{numbers := Numbers}) ->
    lists:sum(Numbers).
In this case, command spec should define handler either as fun, or as a tuple
{module(), atom()}, this requires function to be exported:
%% map form:
cli() -> #{commands => {"sum" => #{handler => fun sum/1}}}.

%% exported function, map form:
cli() -> #{commands => {"sum" => #{handler => {?MODULE, sum}}}}.
Handler may accept positional arguments:
start(Server, Mode) ->
    supervisor:start_child(my_server_sup,
        #{id => Server, start => {my_server, start_link, [Mode]}}).
In this case handler specification must define default value for missing arguments:
handler => {fun start/2, undefined}
%% ... or, for exported function:
handler => {?MODULE, start, undefined}
Default handler
When handler is not specified, cli generates default one, based on the module
implementing cli behaviour, and default field passed to run/2. When
the field it set, positional form for handler is generated, with default set to
the term supplied.
cli does not check whether handler function exists or exported.
Help/usage information
By default, when parser encounters an error, cli generates both error message and help/usage
information. It is possible to suppress usage messages by providing help option
set to false. This also disables usage output for --help or -h request.
As most escripts are interpreted via erl, it is recommended to define progname
to provide correct help/usage line:
cli:run(["arg"], #{progname => "mycli"}).
Reference
cli is able to pass prefixes option to argparse (this also changes -h and --help
prefix). There are also additional options to explore, see cli:run/2 function reference.
cli also passes default option to argparse.



  

    
argparse reference
    

Parser operates with arguments and commands, organised in a hierarchy. It is possible
to define multiple commands, or none. Parsing always starts with root command,
named after init:get_argument(progname). Empty command produces empty argument map:
1> parse("", #{}).
#{}
Options specification
It's possible to override program name using progname option:
2> io:format(args:help(#{}, #{progname => "readme"})).
usage: readme
To override default optional argument prefix (-), use prefixes option:
3> args:parse(["+sbwt"], #{arguments => [#{name => mode, short => $s}]}, #{prefixes => "+"}).
#{mode => "bwt"}
To define a global default for arguments that are not required, use default option:
4> args:parse([], #{arguments => [#{name => mode, required => false}]}, #{default => undef}).
#{mode => undef}
When global default is not set, resulting argument map does not include keys for arguments
that are not specified in the command line and there is no locally defined default value.
Validation, help & usage information
Function validate/1 may be used to validate command with all sub-commands
and options without actual parsing done.
4> args:validate(#{arguments => [#{short => $4}]}).
** exception error: {args,{invalid_option,["erl"],
    [],name,"argument must be a map, and specify 'name'"}}
Human-readable errors and usage information is accessible via help/2 and format_error/1,2.
Return value
If root level command does not contain any sub-commands, parser returns plain map of
argument names to their values:
3> args:parse(["value"], #{arguments => [#{name => arg}]}).
#{arg => "value"}
This map contains all arguments matching command line passed, initialised with
corresponding values. If argument is omitted, but default value is specified for it,
it is added to the map. When no local default value specified, and argument is not
present, corresponding key is not present in the map, unless there is a global default
passed with parse/3 options.
Missing required (field required is set to true for optional arguments,
or missing for positional) arguments raises an error.
When there are sub-commands, parser returns argument map, deepest matched command
name, and a sub-spec passed for this command:
4> Cmd =  #{arguments => [#{name => arg}]}.
#{arguments => [#{name => arg}]}
5> args:parse(["cmd", "value"], #{commands => #{"cmd" => Cmd}}).
{#{arg => "value"},{"cmd",#{arguments => [#{name => arg}]}}}
Command specification
Command specification may contain following fields:
	commands - sub-commands, name to nested command spec
	arguments - list of argument specifications
	help - string to generate usage information, can be set to hidden, causing this command to be omitted in usage output
	handler - function expected to process this command, or optional atom
	user-defined fields

Missing handler field is automatically populated by CLI framework, when a module
exporting cli/0 also exports function function with arity 1, named
after command:
cli() -> #{commands => #{"run" => #{...}}}.

run(ArgMap) -> ....
If command contains sub-commands, and handler is not present, it is an error
if arguments supplied to parse do not select one of the sub-commands. Set
handler to optional to not require sub-command selected.
Command is matched as a positional argument, taking precedence over other
positional arguments. For any positional argument found, parser will first
attempt to match a sub-command in currently evaluated command. If there is no
match, parser considers next argument as positional. Example:
1> Cmd = #{
    commands => #{sub => #{}},
    arguments => [#{name => positional}]
},
2> parse(["arg", "sub"], Cmd) == parse(["arg", "sub"], Cmd).
true
Command names cannot start with prefix character.
Argument specification
Every argument spec must have name field. Name defines key in the parsed
arguments map, similar to dest field of Python library. It is allowed to have
multiple arguments with the same name (option aliases):
    options => [
        #{name => foo, short => $f},
        #{name => foo, long => "-foo"},
        #{name => foo}
    ].
An argument can be:
	optional: matching command line arguments with prefix character
	positional: arguments not starting with a prefix

Negative number is considered positional argument, when - prefix is used, and
there is no no optional argument spec that has short or long form
defined as number (#{name => one, short => $1}).
Argument is optional when short or long field is defined. Short form may only
contain a single character, and long form may contain any string. A single prefix
character is automatically prepended to long forms, so for this example:
    #{name => myarg, short => $m, long => "-myarg"}
Argument myarg can be specified as -m or as --myarg. Please note that it
is possible to have long forms with a single prefix character:
    #{name => kernel, long => "kernel"}
By default, optional arguments are not required and may be omitted from command line.
Positional arguments are required. It is possible to override this behaviour by
setting required field.
It is supported to specify long-form value using this syntax:
    mycli --arg=value

It is treated the same way as mycli --arg value.
For every argument matched, parser may consume several following positional arguments
(not starting with a prefix). Analysis is based on nargs, action and type fields.
	when nargs is 'maybe' and next argument is positional, it gets consumed and produced
 as value, and if next argument starts with option prefix, default value is produced
 (when no default present, default value deduced from argument type)
	when nargs is {'maybe', Term}, and next argument starts with option prefix, Term is
 produced
	when nargs is set to a positive integer, parser tries to consume exactly this
 number of positional arguments, and fails with an error if there is not enough.
 Produced value is a list.
	when nargs is set to nonempty_list, parser consumes at least one following positional
 argument (fails if first argument is not positional), until it finds next optional
 argument. Produced value is a list.
	when nargs is list, parser consumes positional arguments until it finds next
 optional. Produced value is a list.
	when nargs is all, all remaining arguments (whether positional or not) are folded
 into single produced list.

When nargs is not specified, action determines how many arguments are consumed. For
store/append, single argument is consumed. Special handling is done for boolean type:
no argument is consumed, even for store/append. If it is necessary to consume an argument
for boolean type, use type => atom with choices => ["true", "false"].
Action can have following values:
	store - replace value in the argument map (last value wins)
	append - append value to existing list (when value is also a list, resulting
value can be a list of lists)
	{store, Term} - do not consume argument, replace current value with Term
	{append, Term} - do not consume argument, append Term to list of existing values
	count - do not consume any arguments, bump the counter - useful for bumping logging
verbosity level, e.g. cmd -vvv
	extend - valid only for nargs set to list/nonempty_list/all/(pos_integer()), explodes
this list and appends every single value to existing list, so result is just a list, not
a list of lists

Arguments have type associated. Default type is string, taken unprocessed from command
line input.
	int, {int, Limits} - where Limits is a list containing {min, Int}, {max, Int} tuples
	float, {float, Limits} - same as int, but for floating point
	string, {string, Regex}, {string, Regex, RegexOption} - string validated with Regex
	binary, {binary, Regex}, {binary, Regex, RegexOption} - same as string, but of binary type
	atom - existing atom (error when atom does not exist)
	{atom, unsafe} - atom, creates new atoms when needed
	{custom, Fun} - custom type conversion from string() into term()

For int, float, string, binary and atom, it is also possible to specify list of
choices that will pass validation:
    #{name => choices, short => $c, type => {string, ["one", "two", "three"]}}
Custom function may throw erlang:error(invalid_argument), to utilise built-in
validation information. If any other exception is raised, it is passed with no conversion.
An error is thrown when argument cannot be converted to required type, or does not pass validation.
If custom conversion function throws error(invalid_argument), exception is augmented with necessary
parser state information (any other exception passes through unchanged).
It is not allowed to store user-defined fields in arguments.
Errors
Argparse throws exceptions of class error, and Reason tuple:
    {args, ArgParseError}
To get human-readable representation:
    try args:parse(Args, Command)
    catch error:{args, Reason} ->
        io:format(args:format_error(Reason, Command, #{}))
    end.
Help templates
To completely hide the argument or command from the output, supply help => hidden.
It is possible to override help text generated for arguments. By default,
options are formatted with "help text, type, default", e.g.:
crawl [-s <shard>...] [-z <last>] [-c <choice>]

Optional arguments:
  -s     initial shards, int
  -z     last shard, between, 100 < int < 200, 150
  -c     tough choice, choice: 1, 2, 3
It is possible to override the description, and print it this way:
crawl [-s SHARD] [-c CHOICE]

Optional arguments:
  -s     initial number, int, with a default value of 0
  -z     150, number of last shard, unless overridden
  -c     custom parameter, to choose from 1, 2 or 3
Example:
  #{
    arguments => [
      #{
        name => shard,
        default => 0,
        help => {"[-s SHARD]", ["initial number, ", type, " with a default value of ", default]}}
    ]}
First element of the tuple replaces [-s <shard>] with [-s SHARD] in command line example, and
second defines detailed help template.



  

    
Changelog
    

Version 2.0.0
	renamed argparse module to args, to avoid name clash with OTP 26

Version 1.2.4:
	minor spec fixes

Version 1.2.3:
	implemented global default
	minor bugfixes

Version 1.2.1:
	minor bugfixes, support for choices of atoms

Version 1.2.0:
	CLI incompatible change: cli:run/1,2 by default now calls halt(1) in case of a parser error
	bugfixes

Version 1.1.3:
	added help => hidden for commands and options
	changed default formatting for better readability
	fixed bug causing nargs => all to be ignored

Version 1.1.2:
	support for "--arg=value" form

Version 1.1.1:
	added templates for help text

Version 1.1.0:
	Handler support for minimal CLI
	cli/1 optional behaviour callback deprecated
	Ability to provide progname as an atom

Version 1.0.2:
	Improved documentation
	Validation run for CLI commands

Version 1.0.0:
	First public release

Version 0.1.0:
	First usable version (API is unstable yet)




  

    
License
    


BSD 3-Clause License

Copyright (c) 2020-2023, Maxim Fedorov
All rights reserved.

Redistribution and use in source and binary forms, with or without
modification, are permitted provided that the following conditions are met:

1. Redistributions of source code must retain the above copyright notice, this
   list of conditions and the following disclaimer.

2. Redistributions in binary form must reproduce the above copyright notice,
   this list of conditions and the following disclaimer in the documentation
   and/or other materials provided with the distribution.

3. Neither the name of the copyright holder nor the names of its
   contributors may be used to endorse or promote products derived from
   this software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE
DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE LIABLE
FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL
DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR
SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER
CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY,
OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE
OF THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.




  

    
simple
    


#!/usr/bin/env escript

%% simple cli using cli behaviour

-behaviour(cli).
-mode(compile).
-export([cli/0, rm/3]).

main(Args) ->
    cli:run(Args, #{progname => "simple"}).

cli() ->
    #{
        handler => {?MODULE, rm, undefined},
        arguments => [
            #{name => force, short => $f, type => boolean, default => false},
            #{name => recursive, short => $r, type => boolean, default => false},
            #{name => dir}
        ]
    }.

rm(Force, Recursive, Dir) ->
    io:format("Removing ~s (force: ~s, recursive: ~s)~n",
        [Dir, Force, Recursive]).




  

    
calculator
    


#!/usr/bin/env escript

%% calculator, demonstrating sub-commands, and handler specifications
%% how to run from:
%% ERL_FLAGS="-pa ../../../_build/default/lib/argparse/ebin" ./calc mul 2 2

-behaviour(cli).
-mode(compile).
-export([cli/0, sum/1, cos/1, mul/1]).

main(Args) ->
    Out = cli:run(Args, #{progname => "calc"}),
    io:format("~p~n", [Out]).

cli() ->
    #{
        commands => #{
            "sum" => #{
                arguments => [
                    #{name => num, nargs => nonempty_list, type => int, help => "Numbers to sum"}
                ]
            },
            "math" => #{
                commands => #{
                    "sin" => #{handler => {math, sin, undefined}},
                    "cos" => #{},
                    "tan" => #{handler => {math, tan, undefined}}
                },
                arguments => [
                    #{name => in, type => float, help => "Input value"}
                ]
            },
            "mul" => #{
                arguments => [
                    #{name => left, type => int},
                    #{name => right, type => int}
                ]
            }
        }
    }.

sum(#{num := Nums}) ->
    lists:sum(Nums).

cos(#{in := In}) ->
    math:cos(In).

mul(#{left := Left, right := Right}) ->
    Left * Right.




  

    
erm
    


#!/usr/bin/env escript

%% argparse example, without cli behaviour used
%% how to try:
%%  ERL_FLAGS="-pa ../../../_build/default/lib/argparse/ebin" ./erm -rf dir

main(Args) ->
    #{force := Force, recursive := Recursive, dir := Dir} =
        args:parse(Args, cli()),
    io:format("Removing ~s (force: ~s, recursive: ~s)~n",
        [Dir, Force, Recursive]).

%% parser specification
cli() ->
    #{arguments => [
        #{name => force, short => $f, type => boolean, default => false},
        #{name => recursive, short => $r, type => boolean, default => false},
        #{name => dir}
    ]}.




  

    
args 
    



      
Command line parser, made with hierarchy of commands in mind. Parser operates with arguments and commands, organised in a hierarchy. It is possible to define multiple commands, or none. Parsing always starts with root command, named after init:get_argument(progname). Empty command produces empty argument map:   1> parse("", #{}).
      #{}
If root level command does not contain any sub-commands, parser returns plain map of argument names to their values:   3> args:parse(["value"], #{arguments => [#{name => arg}]}).
   #{arg => "value"}
This map contains all arguments matching command line passed, initialised with corresponding values. If argument is omitted, but default value is specified for it, it is added to the map. When no default value specified, and argument is not present, corresponding key is not present in the map.
Missing required (field required is set to true for optional arguments, or missing for positional) arguments raises an error.
When there are sub-commands, parser returns argument map, deepest matched command name, and a sub-spec passed for this command:   4> Cmd =  #{arguments => [#{name => arg}]}.
   #{arguments => [#{name => arg}]}
   5> args:parse(["cmd", "value"], #{commands => #{"cmd" => Cmd}}).
   {#{arg => "value"},{"cmd",#{arguments => [#{name => arg}]}}}

    Types
  


    
      
        arg_map/0

      


        Arguments map: argument name to a term, produced by parser. Supplied to command handler


    


    
      
        arg_type/0

      


    


    
      
        argparse_reason/0

      


    


    
      
        argument/0

      


    


    
      
        argument_help/0

      


    


    
      
        cmd_path/0

      


        Command path, for deeply nested sub-commands


    


    
      
        command/0

      


    


    
      
        command_map/0

      


        Sub-commands are arranged into maps (cannot start with prefix)


    


    
      
        command_spec/0

      


        Command name with command spec


    


    
      
        handler/0

      


    


    
      
        parse_result/0

      


        Result returned from parse/2,3: can be only argument map, or argument map with command_spec.


    


    
      
        parser_options/0

      


    





  
    Functions
  


    
      
        format_error(Reason)

      


        Format exception reasons produced by parse/2. Exception of class error with reason {args, Reason} is normally raised, and format_error accepts only the Reason part, leaving other exceptions that do not belong to argparse out.


    


    
      
        format_error(Reason, Command, Options)

      


        Formats exception, and adds command usage information for command that was known/parsed when exception was raised.


    


    
      
        help(Command)

      


        Equivalent to help(Command, #{}).



    


    
      
        help(Command, Options)

      


        Returns help for Command formatted according to Options specified


    


    
      
        parse(Args, Command)

      


        Equivalent to parse(Args, Command, #{}).



    


    
      
        parse(Args, Command, Options)

      


        Parses supplied arguments according to expected command definition.


    


    
      
        validate(Command)

      


        Equivalent to validate(Command, #{}).



    


    
      
        validate(Command, Options)

      


        Validate command specification, taking Options into account. Generates error signal when command specification is invalid.

      Link to this type
    
    arg_map/0


      
       
       View Source
     


  


  

      

          -type arg_map() :: #{term() => term()}.


      


Arguments map: argument name to a term, produced by parser. Supplied to command handler
  



  
    
      
      Link to this type
    
    arg_type/0


      
       
       View Source
     


  


  

      

          -type arg_type() ::
    boolean | float |
    {float, [float()]} |
    {float, [{min, float()} | {max, float()}]} |
    int |
    {int, [integer()]} |
    {int, [{min, integer()} | {max, integer()}]} |
    string |
    {string, [string()]} |
    {string, string()} |
    {string, string(), [term()]} |
    binary |
    {binary, [binary()]} |
    {binary, binary()} |
    {binary, binary(), [term()]} |
    atom |
    {atom, [atom()]} |
    {atom, unsafe} |
    {custom, fun((string()) -> term())}.


      



  



  
    
      
      Link to this type
    
    argparse_reason/0


      
       
       View Source
     


  


  

      

          -type argparse_reason() ::
    {invalid_command, cmd_path(), Field :: atom(), Reason :: string()} |
    {invalid_option, cmd_path(), Name :: string(), Field :: atom(), Reason :: string()} |
    {unknown_argument, cmd_path(), Argument :: string()} |
    {missing_argument, cmd_path(), argument()} |
    {invalid_argument, cmd_path(), argument(), Argument :: string()}.


      



  



  
    
      
      Link to this type
    
    argument/0


      
       
       View Source
     


  


  

      

          -type argument() ::
    #{name := atom() | string() | binary(),
      short => char(),
      long => string(),
      required => boolean(),
      default => term(),
      type => arg_type(),
      action => store | {store, term()} | append | {append, term()} | count | extend,
      nargs => pos_integer() | maybe | {maybe, term()} | list | nonempty_list | all,
      help => hidden | string() | argument_help()}.


      



  



  
    
      
      Link to this type
    
    argument_help/0


      
       
       View Source
     


  


  

      

          -type argument_help() :: {string(), [string() | type | default]}.


      



  



  
    
      
      Link to this type
    
    cmd_path/0


      
       
       View Source
     


  


  

      

          -type cmd_path() :: [string()].


      


Command path, for deeply nested sub-commands
  



  
    
      
      Link to this type
    
    command/0


      
       
       View Source
     


  


  

      

          -type command() ::
    #{commands => command_map(),
      arguments => [argument()],
      help => hidden | string(),
      handler => handler()}.


      



  



  
    
      
      Link to this type
    
    command_map/0


      
       
       View Source
     


  


  

      

          -type command_map() :: #{string() => command()}.


      


Sub-commands are arranged into maps (cannot start with prefix)
  



  
    
      
      Link to this type
    
    command_spec/0


      
       
       View Source
     


  


  

      

          -type command_spec() :: {Name :: [string()], command()}.


      


Command name with command spec
  



  
    
      
      Link to this type
    
    handler/0


      
       
       View Source
     


  


  

      

          -type handler() ::
    optional |
    fun((arg_map()) -> term()) |
    {module(), Fn :: atom()} |
    {fun((arg_map()) -> term()), term()} |
    {module(), atom(), term()}.


      



  



  
    
      
      Link to this type
    
    parse_result/0


      
       
       View Source
     


  


  

      

          -type parse_result() :: arg_map() | {arg_map(), command_spec()}.


      


Result returned from parse/2,3: can be only argument map, or argument map with command_spec.
  



  
    
      
      Link to this type
    
    parser_options/0


      
       
       View Source
     


  


  

      

          -type parser_options() ::
    #{prefixes => [integer()],
      default => term(),
      progname => string() | atom(),
      command => [string()]}.


      



  


        

      

      
        
          
            
            Anchor for this section
          
Functions
        

        


  
    
      
      Link to this function
    
    format_error(Reason)


      
       
       View Source
     


  


  

      

          -spec format_error(Reason :: argparse_reason()) -> string().


      


Format exception reasons produced by parse/2. Exception of class error with reason {args, Reason} is normally raised, and format_error accepts only the Reason part, leaving other exceptions that do not belong to argparse out.
  



  
    
      
      Link to this function
    
    format_error(Reason, Command, Options)


      
       
       View Source
     


  


  

      

          -spec format_error(argparse_reason(), command() | command_spec(), parser_options()) -> string().


      


Formats exception, and adds command usage information for command that was known/parsed when exception was raised.
  



  
    
      
      Link to this function
    
    help(Command)


      
       
       View Source
     


  


  

      

          -spec help(command() | command_spec()) -> string().


      


Equivalent to help(Command, #{}).

  



  
    
      
      Link to this function
    
    help(Command, Options)


      
       
       View Source
     


  


  

      

          -spec help(command() | command_spec(), parser_options()) -> string().


      


Returns help for Command formatted according to Options specified
  



  
    
      
      Link to this function
    
    parse(Args, Command)


      
       
       View Source
     


  


  

      

          -spec parse(Args :: [string()], command() | command_spec()) -> parse_result().


      


Equivalent to parse(Args, Command, #{}).

  



  
    
      
      Link to this function
    
    parse(Args, Command, Options)


      
       
       View Source
     


  


  

      

          -spec parse(Args :: [string()], command() | command_spec(), Options :: parser_options()) ->
         parse_result().


      


Parses supplied arguments according to expected command definition.
  



  
    
      
      Link to this function
    
    validate(Command)


      
       
       View Source
     


  


  

      

          -spec validate(command()) -> {Progname :: string(), command()}.


      


Equivalent to validate(Command, #{}).
Built-in types include basic validation abilities String and binary validation may use regex match (ignoring captured value). For float, int, string, binary and atom type, it is possible to specify available choices instead of regex/min/max.
  



  
    
      
      Link to this function
    
    validate(Command, Options)


      
       
       View Source
     


  


  

      

          -spec validate(command(), parser_options()) -> {Progname :: string(), command()}.


      


Validate command specification, taking Options into account. Generates error signal when command specification is invalid.
  


        

      



  

    
cli behaviour
    



      
Command line utility behaviour. Usage example:
From an escript main/1 function (requires -mode(compile)):       cli:run(Args).
Or, to limit cli behaviour discovery,       cli:run(Args, #{modules => ?MODULE, progname => ?MODULE}).
Other options available for run/2:	modules:	all_loaded - search all loaded modules (code:all_loaded/0) for cli behaviour
	module() - use this module (must export cli/0)
	[module()] - list of modules (must export cli/0)


	warn: set to suppress suppresses warnings logged
	error: defines what action is taken upon parser error. Use ok to completely ignore the error (historical behaviour, useful for testing), error to raise an exception, halt to halt the emulator with exit code 1 (default behaviour), and {halt, non_neg_integer()} for a custom exit code halting the emulator
	help: set to false suppresses printing usage when parser produces an error, and disables default --help/-h behaviour
	prefixes: prefixes passed to argparse
	progname: specifies executable name instead of 'erl'

Warnings are printed to OTP logger, unless suppressed.
cli framework attempts to create a handler for each command exported, including intermediate (non-leaf) commands, if it can find function exported with suitable signature.
cli examples are available on GitHub
      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Types
  


    
      
        run_options/0

      


    





  
    Callbacks
  


    
      
        cli/0

      


    





  
    Functions
  


    
      
        run(Args)

      


        Equivalent to run(Args, #{}).



    


    
      
        run(Args, Options)

      


        CLI entry point, parses arguments and executes selected function. Finds all modules loaded, and implementing cli behaviour, then matches a command and runs handler defined for a command.


    





      


      
        
          
            
            Anchor for this section
          
Types
        

        


  
    
      
      Link to this type
    
    run_options/0


      
       
       View Source
     


  


  

      

          -type run_options() ::
    #{modules => all_loaded | module() | [module()],
      warn => suppress | warn,
      help => boolean(),
      error => ok | error | halt | {halt, non_neg_integer()},
      prefixes => [integer()],
      default => term(),
      progname => string() | atom()}.


      



  


        

      

      
        
          
            
            Anchor for this section
          
Callbacks
        

        


  
    
      
      Link to this callback
    
    cli/0


      
       
       View Source
     


  


  

      

          -callback cli() -> args:command().


      



  


        

      

      
        
          
            
            Anchor for this section
          
Functions
        

        


  
    
      
      Link to this function
    
    run(Args)


      
       
       View Source
     


  


  

      

          -spec run(Args :: [string()]) -> term().


      


Equivalent to run(Args, #{}).

  



  
    
      
      Link to this function
    
    run(Args, Options)


      
       
       View Source
     


  


  

      

          -spec run([string()], run_options()) -> term().


      


CLI entry point, parses arguments and executes selected function. Finds all modules loaded, and implementing cli behaviour, then matches a command and runs handler defined for a command.

